

# NAG Library Routine Document

## F07MEF (DSYTRS)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07MEF (DSYTRS) solves a real symmetric indefinite system of linear equations with multiple right-hand sides,

$$AX = B,$$

where  $A$  has been factorized by F07MDF (DSYTRF).

### 2 Specification

```
SUBROUTINE F07MEF (UPLO, N, NRHS, A, LDA, IPIV, B, LDB, INFO)
INTEGER          N, NRHS, LDA, IPIV(*), LDB, INFO
REAL (KIND=nag_wp) A(LDA,*), B(LDB,*)
CHARACTER(1)     UPLO
```

The routine may be called by its LAPACK name *dsytrs*.

### 3 Description

F07MEF (DSYTRS) is used to solve a real symmetric indefinite system of linear equations  $AX = B$ , this routine must be preceded by a call to F07MDF (DSYTRF) which computes the Bunch–Kaufman factorization of  $A$ .

If  $UPLO = 'U'$ ,  $A = PUDU^T P^T$ , where  $P$  is a permutation matrix,  $U$  is an upper triangular matrix and  $D$  is a symmetric block diagonal matrix with 1 by 1 and 2 by 2 blocks; the solution  $X$  is computed by solving  $PUDY = B$  and then  $U^T P^T X = Y$ .

If  $UPLO = 'L'$ ,  $A = PLDL^T P^T$ , where  $L$  is a lower triangular matrix; the solution  $X$  is computed by solving  $PLDY = B$  and then  $L^T P^T X = Y$ .

### 4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

1: UPLO – CHARACTER(1) *Input*

*On entry:* specifies how  $A$  has been factorized.

UPLO = 'U'  
 $A = PUDU^T P^T$ , where  $U$  is upper triangular.

UPLO = 'L'  
 $A = PLDL^T P^T$ , where  $L$  is lower triangular.

*Constraint:* UPLO = 'U' or 'L'.

- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides.  
*Constraint:*  $NRHS \geq 0$ .
- 4: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* details of the factorization of  $A$ , as returned by F07MDF (DSYTRF).
- 5: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07MEF (DSYTRS) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 6: IPIV(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .  
*On entry:* details of the interchanges and the block structure of  $D$ , as returned by F07MDF (DSYTRF).
- 7: B(LDB,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array  $B$  must be at least  $\max(1, NRHS)$ .  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ .  
*On exit:* the  $n$  by  $r$  solution matrix  $X$ .
- 8: LDB – INTEGER *Input*  
*On entry:* the first dimension of the array  $B$  as declared in the (sub)program from which F07MEF (DSYTRS) is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 9: INFO – INTEGER *Output*  
*On exit:*  $INFO = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $INFO = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

For each right-hand side vector  $b$ , the computed solution  $x$  is the exact solution of a perturbed system of equations  $(A + E)x = b$ , where

if UPLO = 'U',  $|E| \leq c(n)\epsilon P|U||D||U^T|P^T$ ;

if UPLO = 'L',  $|E| \leq c(n)\epsilon P|L||D||L^T|P^T$ ,

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

If  $\hat{x}$  is the true solution, then the computed solution  $x$  satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(n) \text{cond}(A, x) \epsilon$$

where  $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty} \leq \text{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$ .

Note that  $\text{cond}(A, x)$  can be much smaller than  $\text{cond}(A)$ .

Forward and backward error bounds can be computed by calling F07MHF (DSYRFS), and an estimate for  $\kappa_{\infty}(A)$  ( $= \kappa_1(A)$ ) can be obtained by calling F07MGF (DSYCON).

## 8 Parallelism and Performance

F07MEF (DSYTRS) is not threaded by NAG in any implementation.

F07MEF (DSYTRS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $2n^2r$ .

This routine may be followed by a call to F07MHF (DSYRFS) to refine the solution and return an error estimate.

The complex analogues of this routine are F07MSF (ZHETRS) for Hermitian matrices and F07NSF (ZSYTRS) for symmetric matrices.

## 10 Example

This example solves the system of equations  $AX = B$ , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -9.50 & 27.85 \\ -8.38 & 9.90 \\ -6.07 & 19.25 \\ -0.96 & 3.93 \end{pmatrix}.$$

Here  $A$  is symmetric indefinite and must first be factorized by F07MDF (DSYTRF).

### 10.1 Program Text

```

Program f07mefe

!      F07MEF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: dsytrf, dsytrs, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Integer                     :: i, ifail, info, lda, ldb, lwork, n, &
                                   nrhs
      Character (1)               :: uplo

```

```

!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:), work(:)
      Integer, Allocatable           :: ipiv(:)
!      .. Executable Statements ..
      Write (nout,*) 'F07MEF Example Program Results'
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, nrhs
      lda = n
      ldb = n
      lwork = 64*n
      Allocate (a(lda,n),b(ldb,nrhs),work(lwork),ipiv(n))

!      Read A and B from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Read (nin,*)(a(i,i:n),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)(a(i,1:i),i=1,n)
      End If
      Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Factorize A
!      The NAG name equivalent of dsytrf is f07mdf
      Call dsytrf(uplo,n,a,lda,ipiv,work,lwork,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!      Compute solution
!      The NAG name equivalent of dsytrs is f07mef
      Call dsytrs(uplo,n,nrhs,a,lda,ipiv,b,ldb,info)

!      Print solution

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,nrhs,b,ldb,'Solution(s)',ifail)

      Else
        Write (nout,*) 'The factor D is singular'
      End If

      End Program f07mefe

```

## 10.2 Program Data

F07MEF Example Program Data

```

  4  2          :Values of N and NRHS
  'L'          :Value of UPLO
  2.07
  3.87 -0.21
  4.20  1.87  1.15
-1.15  0.63  2.06 -1.81  :End of matrix A
-9.50 27.85
-8.38  9.90
-6.07 19.25
-0.96  3.93          :End of matrix B

```

### 10.3 Program Results

F07MEF Example Program Results

Solution(s)

	1	2
1	-4.0000	1.0000
2	-1.0000	4.0000
3	2.0000	3.0000
4	5.0000	2.0000

---